
Osvětlovací modul pro RC modely s procesorem PICAXE 08M

Potřebné součástky:
Modul:
1x PICAXE 08M
1x kondenzátor 100nF 1x R 10K
2x R 100R ***
2x R 270R
1x R 220R
1x lámací lišta jednořadá 90° 2.54mm (3 piny v kuse)
1x lámací lišta dvouřadá 90° 2.54mm (6 dvojic v kuse)
1x servokabel JR/Futaba (samec) 15cm
materiál k výrobě DPS (univerzální vrtaná deska min. 13x7 otvorů nebo cuprexit)

***) předřadné odpory. Možno nahradit jinými podle použitých LED

Světla:
5x LED dioda (3xčervená, 1x zelená, 1xmodrá nebo bílá )
6x konektor JR/Futaba (samec) nebo jiný vhodný k lámací liště
1x tahový přepínač nebo zkratovací propojka
dvoužilový kabel příslušné délky k vyvedení světel

Programování:
seriový programovací kabel k programování procesorů PICAXE
1x R 10K
1x R 22K
2x lámací lišta jednořadá 90° 2.54mm (3 piny v kuse)
univerzální vrtaná deska 5x6 otvorů
programovací sofware (např: PICAXE Programing Editor)



Program 1: (příklad ovládání funkce pomocí signálu z přijímače)

;------------ PICAXE 08M – Modul osvětlení pro RC modely ------------
;------------            (c)2009 Milan Žejdlík           ------------
symbol delkapulsu = 150 ;nastavení konstant
symbol minimum = 0                        
symbol maximum = 20
Init:

for b0 = 0 to 2 ;Po zapojení světla 3x bliknou (kontrola)
   switch on 0
   switch on 2
   switch on 4
   pause 300
   switch off 0
   switch off 1
   switch off 2
   switch off 4
   pause 300

next b0
main:
  switch on 4 ;navigační světla vždy svítí
Check:
  pulsin 3,1,w6 ;čtení délky pulzu z přijímače
  if w6>=delkapulsu then gosub LightsON ;puls >= než 1,5ms = STROBE a BEACON zapnuto 
   if w6<delkapulsu then goto Check ;puls < než 1,5ms = STROBE a BEACON vypnuto 
   goto Check
LightsON: ;STROBE 3x blikne, potom jeden cyklus Beacon

gosub Flash
   pause 550

gosub Flash
   pause 550
   gosub Flash
   gosub Beacon
   return  
Flash: ;STROBE light SubRoutine
   switch on 2
   pause 100
   switch off 2
   return
Beacon: ;Efekt pomalu blikajících Beacon majáčků

b2 = maximum ;hodnoty nastaveny v úvodu programu
b4 = minimum
for b1 = minimum to maximum step 1 ;Pozvolné rozsvěcení. Od Min. do Max., krok 1

switch on 0 ;Rozsviť
switch on 1
pause b1 ;pauza1 - prodlužuje se s počtem cyklů
switch off 0 ;Zhasni
switch off 1
pause b2 ;pauza2 - v prvním kroku = Maximum
let b2=b2-1 ;každý další krok se zkracuje o 1 až k Minimum

next b1
;Beacony jsou rozsvíceny na max.výkon, následuje bliknutí Strobů a pozvolné zhasnutí
switch on 0
switch on 1
gosub Flash
for b3 = maximum to minimum step -1 ;Pozvolné zhasínání. Od Max. do Min. krok -1

switch on 0
switch on 1
pause b3 ;pauza1 - zkracuje se s počtem cyklů
switch off 0
switch off 1
pause b4 ;pauza2 - v prvním kroku = Minimum

let b4=b4+1 ;každý další krok se prodlužuje o 1 
next b3
return



Program 2: (příklad napájení z externího zdroje)

;------------ PICAXE 08M – Modul osvětlení pro RC modely ------------
;------------            (c)2009 Milan Žejdlík           ------------
Init:

for b0 = 0 to 2 ;Po zapojení světla 3x bliknou (kontrola)
   switch on 0
   switch on 2
   switch on 4
   pause 300
   switch off 0
   switch off 1
   switch off 2
   switch off 4
   pause 300

next b0
main:
  switch on 4 ;navigační světla vždy svítí

gosub Flash ;STROBE 3x blikne, potom jeden cyklus Beacon
   pause 500

gosub Flash
   pause 500
   gosub Flash
   gosub Beacon
   goto main
Flash: ;STROBE light SubRoutine
   switch on 2
   pause 100
   switch off 2
   return
Beacon: ;Dva rychlé záblesky za sebou

for b1 = 0 to 1 ;První světlo
switch on 1
pause 50
switch off 1
pause 50

   next b1
   for b2 = 0 to 1 ;Druhé světlo

switch on 0
pause 50
switch off 0
pause 50

   next b2
   return

