PROJECT: Motor Control

Level 2 - Exercises 1 to 10 – Basic Motor Control
Level 3 - Exercises 11 to 19 – Forward / Reverse Ramp Speed Control

Objective:

Explore electric motor speed control, control logic,
Pulse Width Modulation – PWM, application examination,
speed graph illustrations, ramp speed control.

Materials:

This project incorporates the use of the PicPatch^{08} microcontroller circuit board with the standard kitset. Have ready a small electric motor 3V to 12Vdc, insulted hook-up wire, sticky tape and blue-tac™.

Motor Driver Device.

The current rating of a microcontroller output pin is 12 to 27mA, which is insufficient to drive the motor directly. A higher current switching device called a driver transistor (NPN BC337) will handle over 200mA.

Connection Diagrams.

The microcontroller driver circuit requires the following components fitted onto the PicPatch^{08}as illustrated in the motor driver circuit diagram Figure 1.
R2b = 330Ω (330 ohm) Resistor,
R2a , R2c and R1d leave vacant.
Solder a wire to each of the two motor terminals with one inserted into T2 +V terminal and the other inserted into Pin 2 terminal.
Solder across the motor terminals a 10nF to 100nF capacitor.
Check the value of the capacitor, it may read 0.01uF to 0.1uF.
The colour of the capacitor may be different from that illustrated.

FIGURE 1. Motor driver circuit.



NOTE:
The “Back EMF” from the motor winding will cause high voltages to appear, therefore ensure D2, the “Free-wheel” diode, is fitted correctly to protect the transistor from over voltage.
To minimise electrical noise interference from the motor brushes a capacitor must be fitted across the motor terminals as illustrated in Figure 1.

Back EMF.



The motor rotor, which is the part that spins, is constructed with coils of wire wrapped around iron. This construction acts as an inductor or electro- magnet where electrical energy is stored in the form of magnetic energy or magnetic flux.
When the electric current is switched off to the electro-magnet the magnetic field around the coil collapses.
As the magnetic field collapses this causes an electric current to continue to flow in the coil of wire, which is called Electro Motive Force or Back EMF.
The Back EMF current will cause a very high voltage, kilovolts, to appear across the switch terminals or in this case across the transistor collector pin and the +5V rail. Providing a current path using a free-wheel diode or clamp diode for the back EMF current will reduce the voltage to the 0.6V, the forward voltage drop across the diode. See D2 in Figure 1 - Motor driver circuit.

Electrical Noise Interference.

To produce the rotation in the rotor the commutator, which is connected to the rotor, and the brushes
form a switch to constantly change the magnetic polarity of the rotor electro-magnets.
This switching action produces high frequency electrical noise or radio waves to radiate out from the wires connected to the motor. This radiated electrical noise can affect nearby radios and the microcontroller so that they will not function properly.
To minimise the effects of this electrical interference a local decoupling capacitor is fitted to short circuit these high frequency currents. This decoupling capacitor must be connected as close as possible to the motor terminals for it to be most effective. Do not fit the capacitor at T2 terminals.

Control Interaction.

Fit the following components for the switch circuit as illustrated in Figure 2.

R3d = 10kΩ Resistor, (Pull up)
R3 which is a 330Ω resistor should already be permanently fitted on the circuit board.
For manual control fit switch 3 or you may wish to use a remote micro switch or two wires with bare ends. The remote switch must connect between T3 terminals 0V and Pin3.

FIGURE 2. Switch circuit.



Enter program 1 into the Editor Program and down load into the Picaxe micro.

PROGRAM 1.

This program toggles the motor on or off the instant Switch 3 is **released**.

```
                                    'Motor control On/Off the instant Switch 3 is released.
        Dirs = %00000101            'Set pins 0 and 2 as outputs.
Loop:   Pause 100                   'Wait for 0.1 seconds
        If Pin3 = 1 Then Loop       'If Pin 3 is high (switch 3 not pressed) then loop
NoToggle:
        If Pin3 = 0 Then NoToggle   'If the switch is still pressed cycle until switch 3 is released
        Bit0 = Bit0 ^ 1             'otherwise invert the state of Bit 0 variable
        If Bit0 = 1  Then MotorOn   'If Bit 0 is set to 1 "ON" then switch motor on
        Low 2                       'otherwise switch pin 2 low (Motor Off)
        Goto Loop                   'and repeat the exercise
MotorOn:
        High 2                      'switch pin 2 high (Motor On)
        Goto Loop                   'and repeat the exercise
```

Exercises.

1. Describe an application this technology is commonly used and the benefit for a touch control switch system over an ordinary toggle switch?
2. Modify program 1 so that the motor switches on and off the instant switch 3 is **pressed**.
3. Appliances can have either one "on/off" toggle switch or two separate "On" and "Off" switches. What are the benefits and disadvantages of both systems?

Fit the following components for the Beeper and LED circuit illustrated in Figure 2.

R0a = 330Ω Resistor,
R0b = 330Ω Resistor,
TR0 = Fit an LED into TR0 SIL socket as illustrated.
The Piezo disk connects to T0 terminal Pin 0 and 0V.
The Piezo beeper disk must be fixed on a firm surface for it to produce a sound.
You may wish to use double sided sticky tape or blue-tac.

FIGURE 3. Beeper and LED circuit.



4. Modify program 1 so that the LED will flash while the motor is not operating but glow continually when the motor is operating.
   Make the beeper sound a short tone whenever the button is pressed.
   Hint:
   Use the "Pins" command to control Pin 0 and Pin 2. "Pins = %00000101 or Pins = 0"
5. What added advantage would this feature of the LED flashing and Beeper provide for an elevator in a multi-storey building?

Fit the following components for the switch circuit as illustrated in Figure 4.

R4d = 10kΩ Resistor, (Pull up)
R4a = 330Ω resistor.
For manual control fit switch 4 or you may wish to use a remote micro switch or two wires with bare ends. The remote switch must connect between T4 terminals 0V and Pin4.

FIGURE 4. Additional switch circuit.



Enter program 1 into the Editor Program and down load into the Picaxe micro.

PROGRAM 2.

Pressing Switch 3 switches the motor on, pressing Switch 4 switches the motor off.

```
                                    'Motor control,  Switch 3 = Start, Switch 4 = Stop.
        Dirs = %00000101            'Set pins 0 and 2 as outputs.
Loop:   Pause 100                   'Wait for 0.1 seconds
        If Pin3 = 0 Then MotorOn    'If Pin 3 is high (switch 3 not pressed) then loop
        If Pin4 = 0 Then MotorOff   'If the switch is still pressed cycle until switch 3 is released
        Goto Loop
MotorOn:
        High 2                      'switch pin 2 high (Motor On)
        High 0
        Goto Loop                   'and repeat the exercise
MotorOff:
        Low 2                       'switch pin 2 high (Motor On)
        Low 0
        Goto Loop                   'and repeat the exercise
```

6. Set up the additional switch circuit illustrated above in Figure 4.
   Enter program 2 and down load this into the Picaxe micro.
   Identify what basic safety feature should be observed that is not in this program?

7. If this safety feature was not identified in question 7 and this application was used in an industrial application for operating a conveyor belt explain what problem could occur if "On" switch remained closed?

8. Modify Program 2 so that the Stop button SW4 has priority over the Start button SW3. With the Start button held closed the motor must stop when the Stop button is pressed and remain stopped after the Stop button is released.
   The motor should only start again when the Start button is released and reactivated.

9. Modify Program 2 so that when the motor is started the motor slowly increases to full speed using the PWM command. See Graph 1 - Motor Speed.

   Hint: Insert command lines:

```
For b1 = 20 to 255 Step 2     'Speed increases by 0.78% from 0 to 100%
   PWM 2, b1,10               'output the speed to Pin 2
Next                          'and do the next step
```

Graph 1. Motor Speed.



10. Modify Graph 1 illustrating the motor starting when the Start button SW3 is pressed and the motor stopping when the Stop button SW4 is pressed as in example Question 8.

Exercises for Level 3.

Objective: Develop a motor speed control concept using the PWM control.
Develop a project board for Forward and Reverse Control.

Materials: A small electric motor 3V to 12Vdc, 2x IN4004 or equivalent Schottky diodes, 4x BC546 or BC337 NPN transistors, insulted hook-up wire, sticky tape, blue-tac™, insulated hook up wire, drawing paper and coloured pencils.

PWM Speed Control

Pulse Width Modulation PWM is a switching action where the ratio of on and off periods produces an average voltage. See Graph 2 – Pulse Width Modulation PWM waveforms.
Controlling the average voltage applied to the electric motor will vary the rotating speed of the motor shaft, however this method of control is very dependant on load restrictions on the shaft.

The PWM command produces and output signal illustrated in Graph 2.
The T_{on} period divided by the total T_{on} and T_{off} period λ is the PWM ratio.
The variable b1 in the command line PWM 2, b1,10, produces a PWM ratio b1÷255.
Therefore b1 can be represented as T_{on} and the total period $\lambda = 255$.

Calculating the average applied voltage use the following formula:

$$V_{motor} = \left[\frac{b1}{255} \times V_{supply} \right]$$

Example: If b1 = 100
For a 4.5V battery supply V_{motor} = 100 / 255 x 4.5Vdc = 1.76V
For a 5Vdc supply V_{motor} = 100 / 255 x 5.0Vdc = 1.96V

Graph 2. Pulse Width Modulation PWM waveforms.



11. Develop a program so that pressing Switch 3 starts the motor and switch 4 stops the motor. The motor must slowly increase to maximum speed from start and when Switch 4 has been depressed then slowly decrease to zero speed.
The LED on Pin Port 1 must glow while the motor shaft is spinning.

Hint: Use command lines:

```
For b1 = 20 to 255 Step 2     'Speed increases by 0.78% from 0 to 100%
   PWM 2, b1,10               'output the speed to Pin 2
Next                          'and do the next step

For b1 = 20 to 255 Step -2    'Speed decreases by 0.78% from 100 to 0%
   PWM 2, b1,10               'output the speed to Pin 2
Next                          'and do the next step
```

12. Modify the Program for question 11 so that the stop switch has priority, even while the motor is increasing in speed. The motor must ramp down from the speed it was running the instant the stop button is pressed.

13. **Ramp Speed Control**
Develop a program so that the motor speed slowly increases while pressing switch 3 and slowly decreases in speed while pressing switch 4. Switch 4 must have priority. As soon as the either button is released the motor remains at the set speed.

14. Develop a Graph to illustrate the action of ramp speed control in Question 13. Display on the graph the speed increasing while Switch 3 is depressed and decreasing speed when switch 4 is depressed. Illustrate what happens when both switches are depressed at the same time.

Basic Forward / Reverse Motor Control

The following schematic illustration Figure 5 is a very basic Forward Reverse motor control circuit using a minimum amount of external components for this project.

2 x BC546 or BC337 Transistors 2x IN4004 or equivalent Schottky Diodes.

This circuit uses two Port Pins of the Micro, Pin 1 for forward and Pin2 for reverse.

The P9000 motor supplied with the Standard kit will operate from 1V up to 12Volts dc. This circuit runs very well on 5V but if you decide to use a +12 Volt supply then change the 2x 330R Pull-up resistors to 560R 0.25 Watt or 330R 0.5 Watt.

Note: The circuit will not operate if you have both Pin1 and Pin2 set High.
You will be able to operate this circuit well using the PWM command for speed control,

15. Develop a program to control the speed and direction of the motor with the following requirements;
Depressing Switch 3 decreases the reverse speed and increases the forward speed,
Depressing Switch 4 decreases the forward speed and increases the reverse speed.
With both switches depressed the motor stops and remains stopped after the switches have been released.

16. Draw a graph illustrating the switch action for question 15.

17. Draw a circuit diagram of Figure 5 schematic.
Illustrate on the diagram the back EMF current flow that is experienced when transistor TR1 has just been switched off.
HINT: The top Transistor remains switched on due to the reversed Motor terminal voltage.

Figure 5. A Basic Forward/Reverse Motor Control Schematic Diagram.



18. Explain how you could implement indication using two different coloured LEDs for motor speed direction indication.

19. Describe in detail an application where this type of control system could be used for a crane gantry or model railway train?
What particular type of control switch mechanism would be suitable for this application?

Tutor Information

The first section level 2 demonstrates logic control requirements when working with electric motors and the safety precautions that need to be considered.
Make sure the D2 free wheel diode is fitted before you operate the motor otherwise if the diode is not fitted the BC337 transistor will eventually fail due to over voltage stress.
It is important to fit the decoupling capacitor directly across the motor terminals. If you are using a different motor you may require the larger capacitor value as much as 2020nF or 0.22uF.
If the micro controller circuit works without the motor connected but fails to work with the motor connected due to electrical interference you will need to increase the value of the decoupling capacitor.
If you already have a 100nF capacitor fitted then you may need to increase the value of the Electrolytic Capacitor C4 to 220uF to 470uF.

It is important to consider the applications where an electric motor may be used as some thought will be required to work out the sequence of logic, particular in question 8.
Level 2 has one example of PWM control, however Level 3 has more in-depth information to calculate the duty cycle.
The Forward-Reverse control board project will help the students gain some practical experience in circuit board design and etching. This circuit board will become a major part in future projects.

Answers to Level 2 Exercises

1. This touch control technology is commonly used in domestic appliances such as dish wash drawers, washing machine, clothes dryer and ovens etc.
   The benefit is that the appliance does not instantly start when it is plugged into a power socket until the "ON" switch is pressed. This provides better safety as the appliance can only switch on under direct supervision after power has been restored.

2. **'Motor control On/Off the instant Switch 3 is pressed.**

```
        Dirs = %00000101          'Make Pins 0 and 2 outputs. Make Pin 1, 3 and 4 inputs
Loop:   Pause 100                 'Wait for 0.1 seconds
        If Pin3 = 0 Then Loop     'If Pin 3 is low (switch 3 is pressed) then loop,
NoToggle:
        If Pin3 = 1 Then NoToggle 'If the switch is not pressed cycle until switch 3 is pressed.
        Bit0 = Bit0 ^ 1           'otherwise invert the state of Bit 0 variable
        If Bit0 = 1  Then MotorOn 'If Bit 0 is set to 1 "ON" then switch motor on
        Low 2                     'otherwise switch pin 2 low (Motor Off)
        Goto Loop                 'and repeat the exercise
MotorOn:
        High 2                    'switch pin 2 high (Motor On)
        Goto Loop                 'and repeat the exercise
```

3. The benefit for a single toggle switch is if there is only one switch which generally costs less and provides a tidy appearance. However if the operation status is not immediately obvious the appliance will require some form of status indicator.
   Two separate switches minimise confusion. Industrial equipment generally require a
   a safety mechanism where the power must be actively disconnected by a separate "OFF" or "STOP" switch which has priority over the "START" switch.

4. **'Motor control with LED and beeper**

```
        Dirs = %00000101           'Make Pins 0 and 2 outputs. Make Pin 1, 3 and 4 inputs
Loop:   Pause 100                  'Wait for 0.1 seconds
        If Pin3 = 0 Then Loop      'If Pin 3 is high (switch 3 no pressed) then don't toggle
        Low 0                      'Make sure the Led is switched off.
NoToggle:
        Pause 300                  'Wait 0.2 seconds
        Pulsout 0, 300             'Flash on LED.
        If Pin3 = 1 Then NoToggle  'If the switch is still pressed cycle until switch 3 is released
        Sound 0, (110,50)
        Bit0 = Bit0 ^ 1            'otherwise invert the state of Bit 0 variable
        If Bit0 = 1  Then MotorOn  'If Bit 0 is set to 1 "ON" then switch motor on
        Pins = %00000000           'otherwise switch pin 2 low (Motor and LED Off)
        Goto Loop                  'and repeat the exercise
MotorOn:
        Pins = %00000101           'switch pin 2 high (Motor and LED Off)
        Goto Loop                  'and repeat the exercise
```

5. The flashing LED indicates the power is on and the elevator is ready for operation.
The LED continually glowing indicates the elevator is in use.
The beeper sound is confirmation the button has been activated by the person pressing it.

6. The basic feature is the Stop switch should have priority over the Start switch.
Holding down both start and stop switches simultaneously the motor should not operate.

7. A problem could occur where the start switch or circuit has a short circuit. If pressing the stop switch does not stop the conveyor belt it will continue to operate, which could cause a major incident.

8. Pressing Switch 4 has priority to stop the motor even while Switch 3 remains pressed.
The motor must not start until Switch 3 is released and pressed again.

'Motor control, Switch 4 = priority Stop, Switch 3 = Start.

```
        Dirs = %00000101              'Set pins 0 and 2 as outputs.
CheckStart:
        If Pin3 = 1 Then CheckStop:   'If Pin 3 is high (switch 3 not pressed) then check stop action
StartMotor:
        High 2                        'switch pin 2 High (Motor On)
        High 0                        'switch pin 0 High (LED On)
CheckStop:
        If Pin4 = 1 Then CheckStart:  'If Pin 4 is high (switch 4 not pressed) then check start action
StopMotor:
        Low 2                         'switch pin 2 Low (Motor Off)
        Low 0                         'switch pin 0 Low (LED Off)
        If Pin3 = 0 Then StopMotor    'If Pin 3 remains Low (switch 3 is pressed) then StopMotor
        Goto CheckStart               'otherwise if switch 3 is released check for start action
```

9. **'Motor control, Switch 4 = priority Stop, Switch 3 = Start.**

```
        Dirs = %00000101              'Set pins 0 and 2 as outputs.
CheckStart:
        If Pin3 = 1 Then CheckStop:   'If Pin 3 is high (switch 3 not pressed) then check stop action
StartMotor:
        High 0                        'switch pin 0 High (LED On)
        For b1 = 20 to 255 Step 2     'Speed increases by 0.78% (2÷255) from 0 to 100% speed
            PWM 2, b1,10              'output the speed to Pin 2
        Next                          'and do the next step
        High 2                        'switch pin 2 High (Motor On)
CheckStop:
        If Pin4 = 1 Then CheckStart:  'If Pin 4 is high (switch 4 not pressed) then check start action
StopMotor:
        Pins = 0                      'switch pin 0 and 2 Low (Motor Off, LED Off)
        If Pin3 = 0 Then StopMotor    'If Pin 3 remains Low (switch 3 is pressed) then StopMotor
        Goto CheckStart               'otherwise if switch 3 is released check for start action
```

10. Speed graph of motor starts and stops the instant the Start and Stop buttons are depressed.



Answers to Level 3 Exercises.

11. **'Motor soft start and soft stop control, Switch 4 = Stop, Switch 3 = Start.**

```
        Dirs = %00000101              'Set pins 0 and 2 as outputs.
CheckStart:
        If Pin3 = 1 Then CheckStop:   'If Pin 3 is high (switch 3 not pressed) then check stop action
StartMotor:
        High 0                        'switch pin 0 High (LED On)
        For b1 = 20 to 255 Step 2     'Speed increases by 0.78% (2÷255) from 0 to 100% speed
           PWM 2, b1, 10              'output the speed to Pin 2
        Next                          'and do the next step
        High 2                        'switch pin 2 High (Motor On)
CheckStop:
        If Pin4 = 1 Then CheckStart:  'If Pin 4 is high (switch 4 not pressed) then check start action
        For b1 = 255 to 20 Step -2    'Speed increases by 0.78% (2÷255) from 0 to 100% speed
           PWM 2, b1, 10              'output the speed to Pin 2
        Next                          'and do the next step
StopMotor:
        Low 2                         'switch pin 2 Low (Motor Off)
        Pause 1000           'For high inertia loads increase delay for shaft to stop spinning
        Low 0                'then switch pin 0 Low (LED Off – shaft has stopped spinning).
        If Pin3 = 0 Then StopMotor    'If Pin 3 remains Low (switch 3 is pressed) then StopMotor
        Goto CheckStart               'otherwise if switch 3 is released check for start action
```

12. **'Motor soft start and soft stop control, Switch 4 = priority Stop, Switch 3 = Start.**

```
        Dirs = %00000101              'Set pins 0 and 2 as outputs.
CheckStart:
        If Pin3 = 1 Then CheckStop:   'If Pin 3 is high (switch 3 not pressed) then check stop action
StartMotor:
        High 0                        'switch pin 0 High (LED On)
        For b1 = 20 to 255 Step 2     'Speed increases by 0.78% (2÷255) from 0 to 100% speed
          If Pin4 = 0 Then MotorOff   'If Pin 4 is low (switch 4 is pressed) then stop the motor
          PWM 2, b1, 10               'output the speed to Pin 2
        Next                          'and do the next step
        High 2                        'switch pin 2 High (Motor On)
CheckStop:
        If Pin4 = 1 Then CheckStart:  'If Pin 4 is high (switch 4 not pressed) then check start action
        Let b1 = 255
MotorOff:
        For b0 = b1 to 20 Step -2     'Speed increases by 0.78% (2÷255) from 0 to 100% speed
           PWM 2, b0, 10              'output the speed to Pin 2
        Next                          'and do the next step
StopMotor:
        Low 2                         'switch pin 2 Low (Motor Off)
        Pause 500                     'for high inertia loads increase delay
        Low 0                         'then switch pin 0 Low LED Off, shaft has stopped spinning
        If Pin3 = 0 Then StopMotor    'If Pin 3 remains Low (switch 3 is pressed) then StopMotor
        Goto CheckStart               'otherwise if switch 3 is released check for start action
```

13. Switch ramp speed control.

```
'Switch 3 increases motor speed, Switch 4 has priority and decreases motor speed.
        Dirs = %00000101            'Set pins 0 and 2 as outputs.

MotorStop:
        Let b1 = 20                 'Reset speed variable to 0 speed
        Pins = 0                    'Switch Low all pins, (Motor Off, LED Off)

CheckDec:
        If Pin4 = 1 Then CheckInc   'If Pin 4 is high (Sw 4 is not pressed) check increase action
        b1 = b1 - 2                 'otherwise decrease speed by 0.78%
        Goto UpdateSpd              'then update  the speed

CheckInc:
        IF b1 >= 253 Then Fullspeed 'If speed is above or equal to 252 then full speed
        If  Pin3 = 1  Then UpdateSpd 'otherwise if Pin 3 is high then ouput speed
        b1 = b1 + 2                 'otherwise increase speed by 0.78%
        High 2                      'Set pin 2 high to nudge motor
        Pause 5                             'for a period of 5 milli seconds

UpdateSpd:
        IF b1 <= 20  Then MotorStop 'If speed is set to below or equal to 20 then stop
        High 0                      'Switch high Pin 0 (LED On)
        PWM 2, b1, 10               'operate motor at set speed for 10 cycles
        Goto CheckDec               'repeat the exercise

FullSpeed:
        High 2                      'Set Pin 2 high (Motor On)
        Goto CheckDec
```

14. **Graph illustrating ramp speed control and priority for Switch 4 Stop button.**



15. Switch ramp speed control.

```
'Switch 3 Increases forward and decrease reverse motor speed, Switch 4 is vice versa.
        Dirs = %00000111                'Set pins 0,1 and 2 as outputs.
MotorStop:
        Let b0 = 19                     'Reset speed variable to 0 speed
        Let b1 = 0                      'Reset speed direction to forward
        Pins = 0                        'Switch all pins Low, (Motor Off, LED Off)
        Pause 1000                'wait duration of 1 second to release buttons during stop command
ButtonChk:
        IfPin3 = 0 and Pin4 = 0 Then MotorStop          'If both switches depressed then Stop Motor
        If Pin3 = 0 Then Fwd            'Switch 3 Forward comand
        If Pin4 = 0 Then Rev            'Switch 4 Reverse Comand
        Goto UpdateSpd                  'If no switch action then update  the speed
Fwd:    If b1 = 1 Then Dec              'If in Reverse mode then decrease speed first to zero speed
Inc:    If b0 = 255 Then UpdateSpd      'If at maximum speed then update speed
        Let b0 = b0 + 1                 'Increase speed by 0.39% of supply Voltage  (1/255)
        Goto UpdateSpd                  'otherwise if Pin 3 is high (Sw3 is not pressed) then ouput speed
Rev:    If b1 = 1 Then Inc              'If in Reverse mode then Increase speed
Dec:    If b0 < 20 Then ChangeDir       'If set speed is below 20 then change direction
        Let b0 = b0 – 1                 'Decrease the speed by 0.39% of supply Voltage  (1/255 x V)
UpdateSpd:
        If b0 < 20  Then MotorStop      'If speed is set to below 20 then stop Motor for 1 second
        High 0                          'Switch Pin 0 high (LED On)
        Let b2 = b1 + 1                 'Calculate the I/O Pin to operate Pin 1 (FWD) or Pin 2 (REV).
        High b2                         'Switch high appropriate output
        Pause 5                         'and wait 5 milliseconds to kick start the Motor
        PWM b2, b0, 20                  'operate motor at set speed for 20 cycles (cycles can altered)
        Goto ButtonChk                  'repeat the exercise
ChangeDir:
        Let b0 = 20                     'Set speed to initiate start mode
        Let b1 = b1 ^ 1                 'Toggle the direction indicator 0 = Forward, 1 = Reverse.
        Goto UpdateSpd                  'then update the motor speed
```

16. Graph of Switch ramp speed control for forward and reverse.



17. Circuit Diagram of Driver Circuit illustrating Back EMF current flow after Switching off TR1.



18. Directional Indication can be illustrated by various ways using the following connections:

1. LEDs in series with R1a and R1b, the resistor values would need to be 100R.
2. A series LED and resistor between Pin 1 to +V and Pin 2 to +V.
3. Use Pin 0 or Pin 4 to drive two LEDs;

Program example using Port Pin 0;

Low 0 'to drive LED1,
High 0 'to drive LED 2.

4.5V Battery supply and Port Pin 4;

Low 4 'to drive LED1,
High 4 'to drive LED2,
Input4 'to switch off LEDs



19. This technology is used on a much larger scale for traverse control of cranes for speed and direction, also for raising and lowering of the crane hoist.
Electric Railway systems utilise this technology for forward and reverse speed control.
This specific example is suitable to use for a Model Railway application, however there are other circuit methods which use a single Port Pin for Forward / Reverse Speed control.
A common switch mechanism used in these applications is the Joy-Stick control method.
The hand movement control action of the Joystick should correspond with the direction of movement and speed of the crane gantry, hoist or locomotive engine.
Moving the Joystick handle to the right would move the crane to the right, or moving the Joystick forward will move the train forward etc.